

# 5

## WiiFlash を使ってみよう [ActionScript 編]

---



HOME



WiiFlashを使えば、ActionScriptを用いて WiiRemote を Adobe Flash（以下、Flash と表記）とつなげることができます。それにより、Web サイトなど Flash を使ったコンテンツを WiiRemote で扱うことができるようになります。

この章では WiiFlash 経由で WiiRemote を扱う方法、そして具体的な作例を紹介しましょう。

## 5.1

# WiiFlash とは

WiiFlash は、WiiRemote と Flash アプリケーションをつなぐためのライブラリです。Joa Ebert 氏と Thibault Imbert 氏という 2 人の開発者により開発が進められているプロジェクトです。

**WiiFlash : Wiimote and Flash**

**URL** <http://wiiflash.bytearray.org/>

## WiiFlash の仕組み

WiiFlash を触る前に、WiiFlash の仕組みについて理解しましょう。WiiFlash は、「WiiFlash Server」と「WiiFlash ActionScript API」の 2 つのパートに分けて提供されています。

### ● WiiFlash Server


C++ もしくは、.NET server を利用して作られたバイナリソケットサーバーで、WiiRemote からの情報を Flash が扱える形に変換して通信します。

### ● WiiFlash ActionScript API (SWC component)

WiiFlash Server を使って WiiRemote と通信するための API です。

WiiFlash は、Windows と Macintosh（以下、Mac と表記）両方に対応しています。ただし、

図5-1 WiiFlashの仕組み



Windows と Mac ではできることが違うので注意しましょう（WiiFlash の機能は 166 ページを参照してください）。

## 環境設定

### Flash CS4 のインストール

まずは、一般的な Flash の開発環境である Adobe Flash CS4 をインストールしましょう。ソフトウェアは購入しなくとも、体験版が利用できます。体験版は使用開始日から 1 ヶ月間のみ有効なので、もし気に入ったら購入しましょう。

#### Adobe Flash CS4 Professional

**URL** <http://www.adobe.com/jp/products/flash/>

#### ● Flash CS4 のダウンロード

Adobe サイトのダウンロードページから、「Flash CS4 Professional」→「体験版」を選択します。なお、ダウンロードには Adobe ID が必要になります。手順に従って、Adobe ID を取得しておきましょう。

図 5-2 ダウンロードページ




図 5-3 Adobe IDの取得



### ● Flash CS4のインストール

ダウンロードしたパッケージ (ZIP ファイル) を解凍し、付属のインストールガイド (Readme.txt) を参照の上インストールしてください。

インストールが完了したらアプリケーションを起動して確かめてください。体験版利用の場合にはソフトウェアのライセンスキー確認画面が出ますが、30日以内であれば「体験版として使用する」をクリックすれば、通常版と同じ機能が使えます。

## WiiFlash のインストール

以降では、WiiFlash バージョン 0.4.5 を使って説明します。

なお WiiFlash Server は「Microsoft .NET Framework 3.0 Redistributable Package」が必要になります※1。ダウンロードはこちらから。


### Microsoft ダウンロードセンター

**URL** <http://www.microsoft.com/downloads/details.aspx?FamilyID=10cc340b-f857-4a14-83f5-25634c3bf043&DisplayLang=ja>

### ● WiiFlash のダウンロード

WiiFlash サイトを開き、左側のメニューにある「PAGES」→「Download」を選択します。


図 5-4 WiiFlash (<http://wiiflash.bytearray.org/>)



※1：第 4 章ですでに .NET 開発環境を導入している場合はインストールする必要はありません。

表示される画面から「Download WiiFlash」をクリックし、アーカイブファイル (ZIP 形式) をダウンロードします。

図 5-5 「Download WiiFlash」をクリック




ダウンロードした WiiFlash のアーカイブを展開すると、次のようなフォルダが現れます。

表 5-1 展開したフォルダ (WiiFlash 0.4.5) ファイルの内容

フォルダ名	概要
Servers/WiiFlash Server (latest version)	WiiFlash Serve 本体
Core/api	WiiFlash API (SWC, sources)
Examples	デモファイル (.fla)
Documentation	ドキュメント

図 5-6 展開した WiiFlash のファイル・フォルダ構造



## WiiFlash の基本機能

前述のように、WiiFlash は Windows と Mac の両方に対応していますが、Windows でできても Mac ではできないということもあるので注意しましょう。

表 5-2 WiiFlash の基本機能 (バージョン 0.4.5)

リモコン	系統	機能	Windows	Mac	
WiiRemote	イベント系	接続ができているか	○	○	
		接続ができた	○	○	
		接続が切れた	○	○	
		赤外線 1 点目感知	○		
		赤外線 1 点目消失	○		
		赤外線 2 点目感知	○		
		赤外線 2 点目消失	○		
	ボタン系	十字キー（上下左右）	○	○	
		A ボタン	○	○	
		B ボタン	○	○	
		+ ボタン	○	○	
		- ボタン	○	○	
		HOME ボタン	○	○	
		① ボタン	○	○	
		② ボタン	○	○	
		電源ボタン	×	×	
	加速度センサー	X 軸の加速度	○	○	
		Y 軸の加速度	○	○	
		Z 軸の加速度	○	○	
		Yaw 値	○	○	
		Pitch 値	○	○	
		Roll 値	○	○	
	赤外線	1 点目	○	○	
		2 点目	○	○	
		1 点目の X 値	○	○	
		1 点目の Y 値	○	○	
		2 点目の X 値	○	○	
		2 点目の Y 値	○	○	
	その他	バッテリーレベル	○	○	
		バイブレーション	○	○	
	ヌンチャク	イベント系	接続ができているか	○	○
			接続ができた	○	○
		接続が切れた	○	○	
ボタン系		StickX	○	○	
		StickY	○	○	
		C ボタン ○	○	○	
		Z ボタン ○	○	○	
加速度センサー		X 軸の加速度	○	○	
		Y 軸の加速度	○	○	
		Z 軸の加速度	○	○	
		Yaw 値	○	○	
		Pitch 値	○	○	
	Roll 値	○	○		
WiiBoard		左上の圧力	○		
		右上の圧力	○		
		左下の圧力	○		
		右下の圧力	○		
		重力の中心点	○		



## WiiFlash のデモファイルでテストする

まずは WiiRemote でテストしてみます。WiiRemote を PC の Bluetooth スタックに接続して、WiiFlash Server を立ち上げましょう。

図 5-7 のように「1 Wiimote(s) found」と表示されればつながっています。

図 5-7 正常に接続されている



接続ができていない場合は図 5-8 のように丸い「WF」の部分が赤く表示されます。

図 5-8 接続できていない



次に、ダウンロードしたパッケージの中にある「Examples」フォルダから Wiimote Demo.fla を開きましょう。

このファイルでは、電源ボタンを除く、WiiRemoteのボタンすべてと「Sensor X」「Sensor Y」「Pitch」「Roll」「Yaw」の加速度センサーの値、さらに、「Battery level」を確認することができます。正常に接続されていれば、ボタンを押すことで、対応する画面上のボタン部分が反応します。

図5-9 動作テスト (Wiimote Demo.fla)



このように、サンプルのデモファイルがついているので、楽に開発を行うことができます。確認ができたなら、次の項ではサンプルファイルで実際の作り方を見てみましょう。

## ボタンアクションの取得を使ったサンプル

まずは、サンプル1を開いてください。このサンプルはボタン1つでFlashのアニメーションを再生／停止する簡単なサンプルです。

ライブラリを確認すると、「a」、「circle」と「synchronize」という2つのムービークリップが入っています。

図5-10 ライブラリの中を確認する



「anime\_mc」はアニメーションが入っているムービークリップです。「synchronize」は WiiFlash Server とやり取りをするために必要なファイルです。

**POINT ▶▶▶** 新規で WiiFlash を使用したアプリケーションを作る際には、必ず API が入っている「org」フォルダを fla ファイルと同じディレクトリに入れて作成を開始しましょう。また、ライブラリの synchronize というムービークリップも必要となるので、忘れずにコピーしましょう。

それでは、さっそくサンプルファイル「wiiFlash\_sample fla」のスク립トを見ていきましょう。

## ライブラリのインポート

まず最初に WiiFlash のライブラリをインポートします。Flash の event クラスも後で使うので、あらかじめインポートしましょう。

### コード 5-1 ActionScript WiiFlash のライブラリをインポートする

```
import org.wiiflash.Wiimote;
import org.wiiflash.events.ButtonEvent;
import org.wiiflash.events.WiimoteEvent;
import flash.events.*;
```

WiiRemote をインスタンス化して、使えるようにします。

コード 5-2 **ActionScript** WiiRemote をインスタンス化

```
// WiiRemoteをインスタンス化する  
var myWiimote:Wiimote = new Wiimote();
```

WiiFlash Server に接続する処理は次の部分です。

コード 5-3 **ActionScript** WiiFlash Server に接続

```
// WiiRemoteをWiiFlash Serverに接続する  
myWiimote.connect ();
```

エラーや接続状況などを管理する「synchronize」というムービークリップをインスタンス化し、画面に配置します。

コード 5-4 **ActionScript** synchronize を配置

```
var mySynchronize:Synchronize = new Synchronize();  
mySynchronize.x = (stage.stageWidth - mySynchronize.width) / 2;  
mySynchronize.y = (stage.stageHeight - mySynchronize.height) / 2
```

エラーなどのイベントハンドラを登録します。

コード 5-5 **ActionScript** イベントハンドラの登録

```
myWiimote.addEventListener( Event.CONNECT, onWiimoteConnect );  
myWiimote.addEventListener( IOErrorEvent.IO_ERROR, onWiimoteConnectError );  
myWiimote.addEventListener( Event.CLOSE, onCloseConnection );
```

コード 5-6 **ActionScript** イベント管理

```
function onCloseConnection ( pEvent:Event ):void{  
    mySynchronize.infos_text = "WiiRemoteとの接続が切れました";  
    addChild( mySynchronize );  
}  
  
function onWiimoteConnectError ( pEvent:IOErrorEvent ):void{  
    mySynchronize.infos_txt.text = "正常に接続ができませんでした。  
        WiiFlash Serverが立ち上がっているか確認してください。";
```


[次ページにつづく](#)

```
}  
  
function onWiimoteConnect ( pEvent:Event ):void{  
    removeChild ( mySynchronize );  
    infos_txt.text = "WiiRemoteと正常に接続されました。";  
}
```

## ボタンイベントを使ったアクション

前述のように、このサンプルはボタン 1 つで Flash のアニメーションを再生／停止する簡単なサンプルです。A ボタンを押すと画面上の円（ムービークリップ：anime\_mc）のアニメーションが再生されます。ボタンを離すとアニメーションは停止します。

図5-11 ボタンでFlashのアニメーションを再生／停止



ボタンを使うにはまず、イベントの登録をします。ボタンのイベント登録は基本的に「on」と「off」を設定します。

- **ButtonEvent.CLICK**

A ボタンが押された際に発行されるイベントです。

- **ButtonEvent.CLICK**

A ボタンが離された際に発行されるイベントです。

**コード 5-7    ActionScript    ボタンイベントの登録**

```
myWiimote.addEventListener( ButtonEvent.A_PRESS, onAPressed );  
myWiimote.addEventListener( ButtonEvent.A_RELEASE, onAReleased);
```

A ボタンが押された際に呼ばれるメソッドがこちらです。  
画面に配置された anime\_mc のアニメーションを再生します。

**コード 5-8    ActionScript    A ボタンが押された際に呼ばれるメソッド**

```
function onAPressed ( pEvt:ButtonEvent ):void {  
    anime_mc.play();  
}
```

A ボタンが離された際に呼ばれるメソッドがこちらです。  
画面に配置された anime\_mc のアニメーションを停止します。。

**コード 5-9    ActionScript    A ボタンが離された際に呼ばれるメソッド**

```
function onAReleased ( pEvt:ButtonEvent ):void {  
    anime_mc.stop();  
}
```

このように WiiFlash では、開発が簡単に始められるように、リモコンのボタンを使うためのメソッドが用意されています。

このサンプルを使って、anime\_mc の再生中身を変更するだけで、簡単に WiiRemote を使った Flash コンテンツを作ることができます。

もっと詳しく勉強したい方は、次節から紹介する作例を参考にしてください。

ここで紹介しているのサンプルの応用や WiiBoard を利用したものになっているので、ぜひ理解して自分なりの作品作りに活かしてください。

## 5.2

## 作例1：座布団 (ザプトン)

制作／尾崎俊介

この節では、WiiBoard と Flash を使ったサンプルを紹介します。

図5-12 座布団 (ザプトン)



Wii コントローラーの中でも特殊な WiiBoard は、4 点の圧力センサーの各箇所ごとの加重を計ることができます※2。この 4 点の合計を取って、体重計を作ることもできます。ほかにも、4 点の圧力センサーのどこに比重が掛かっているかを判断し、上に乗っている人の重心がわかる機能も備えています。バランスを取るゲームでは、こちらの値が使われているのでしょう。今回は、この重心の移動を取得する機能を使ってサンプルを作りました。

※2：WiiBoard の詳しい説明は、「WiiFlash の基本機能」を参考にしてください。

## Flash で姿勢が悪い人を救いたい

姿勢を直すことはとても時間のかかる作業です。姿勢の悪さは、常に気を張って意識をし、その座り方自体を習慣化しないことには解決することはできません。

図 5-13 姿勢を直すことはとてもたいへん

■問題起点

長時間勤務により、姿勢を保つことが困難である。



そこで、この姿勢を直すことを強く「意識する」ということをいかに習慣づけるのかと考えた結果、以下のテーマに落ち着きました。

姿勢が悪いことは恥ずかしい

この「恥ずかしい」という意識こそが、人の姿勢を直す際の大きい助けになると思いました。この恥ずかしいという意識を持ってもらうために開発したサンプルが、姿勢矯正「座布団（ザブトン）」です。



図5-14 座布団 (ザブトン)



この座布団を Bluetooth を使い、PC に接続します。WiiFlash を立ち上げて、認識されたら、Flash アプリを立ち上げます。接続が確認できたら、座布団を普段使っているいすの上に乗せ、その上に座ります。姿勢が重心から外れると「ぷ〜!」とおならの音が鳴ります。座布団の中に WiiBoard を仕込んで、重心移動を測定しているのです。

図5-15 座布団 (ザブトン) の構成





図 5-16 姿勢が重心から外れると「ぶ〜!」



## 重心移動の変化量を使ったアプリ

画面上では、重心の移動を元に黄色の点の位置を変化させます。プログラム (zaputon fla) を見ていきましょう。

まずは初期設定部分です。前節で解説した通り、WiiFlash ライブラリの読み込み、インスタンス化など、WiiFlash を使うために必要な処理です。

コード 5-10 ActionScript WiiFlash の初期設定

```
import org.wiiflash.Wiimote;
import org.wiiflash.events.ButtonEvent;
import org.wiiflash.events.WiimoteEvent;
import flash.events.Event;

// Wiimoteのインスタンス化
var myWiimote:Wiimote = new Wiimote();
// リモコンの接続確認・エラー用にリスナー登録
myWiimote.addEventListener( Event.CONNECT, onWiimoteConnect );
myWiimote.addEventListener( IOErrorEvent.IO_ERROR, onWiimoteConnectError );
// バランスボードへ接続
myWiimote.addEventListener( WiimoteEvent.UPDATE, onUpdated );
myWiimote.addEventListener( WiimoteEvent.BALANCEBOARD_CONNECT, 〇
```

[次ページにつづく ➤](#)

```

                                onBalanceBoardConnected );
myWiimote.addEventListener( WiimoteEvent.BALANCEBOARD_DISCONNECT,
                                onBalanceBoardDisconnected );

// WiiFlash Serverにリモコンを接続
myWiimote.connect ();
//Synchronizeをインスタンス化
var mySynchronize:Synchronize = new Synchronize();
//Synchronizeをステージに配置
addChild( mySynchronize );
//Synchronizeの位置を画面中心へ
mySynchronize.x = (stage.stageWidth - mySynchronize.width) / 2;
mySynchronize.y = (stage.stageHeight - mySynchronize.height) / 2

```

重心移動の変化量を取得するのは以下の部分です。このUpdated関数がWiiBoardに変化が起るたびに呼び出されます。

#### コード5-11   **ActionScript** Updated 関数

```

function onUpdated ( pEvt:WiimoteEvent ):void {
    //ターゲットの場所を移動
    pointTarget.x = pEvt.target.balanceBoard.centerOfGravity.x * 10 + 90;
    pointTarget.y = pEvt.target.balanceBoard.centerOfGravity.y * 20 + 115;
    //変化量を計測
    checkDiff(pEvt.target.balanceBoard.centerOfGravity.x, pEvt.target.
                                balanceBoard.centerOfGravity.y);
}

```

そして、checkDiff関数でポジションを変更します。

#### コード5-12   **ActionScript** checkDiff 関数

```

function checkDiff(gx, gy):void{
    xPos = gx;
    yPos = gy;
}

```

このサンプルでは、Timer 関数を使って定期的に計算をしています。Updated関数でも、定期的に計算ができるのですが、このUpdated関数は呼び出される頻度がとても高いので、処理が重くなってしまうことがあります。そこで今回は、少し間隔をあけて実行するように、次のTimer関数を使用しました。

コード 5-13   **ActionScript**   Timer 関数

```
/**重心の移動変化を確認する間隔*/  
var delay:uint = 3000;  
/**重心の移動変化を確認する間隔*/  
var repeat:uint = 0;  
/**重心の移動変化を確認する間隔*/  
var myTimer:Timer = new Timer(delay, repeat);  
myTimer.addEventListener("timer", timerHandler);  
myTimer.start();
```

実際に計算している部分はコード 5-14 に示した timerHandler 関数です。


コード 5-14   **ActionScript**   timerHandler 関数

```
function timerHandler(e:TimerEvent):void{  
    //変化量を計測  
    var c:Number = Math.sqrt((xPos*xPos)+(yPos*yPos));  
    //変化量を出力  
    _txt.text = String(Math.floor(c));  
    //変化量が規定値以上であればおならを鳴らす  
    if(c > maxAmount){  
        playFarts();  
    }  
}
```

重心は x 軸と y 軸のポイントからなる位置の値として取得することができます。この x 軸と y 軸の位置が中心点 (0,0) のポイントから直線距離でどのくらい離れているかを計算し、その値を元に姿勢が崩れたかどうかを判断しています。

```
var c:Number = Math.sqrt((xPos*xPos)+(yPos*yPos));
```

図5-17 重心のズレから姿勢を判断



そして、playFarts関数でおならを鳴らしています。音を鳴らすために必要であれば、Flaファイルと同じディレクトリにあるseフォルダをswfと同じ階層に移動してください。

**コード5-15** **ActionScript** playFarts関数

```
/**
 * おならを鳴らす
 */
function playFarts():void{
    //鳴らす音を選択
    var soundName = farts_array[Math.floor(Math.random()*farts_array.length)];
    //音のファイルを取り出す
    var req:URLRequest = new URLRequest("se/"+soundName);
    //サウンドオブジェクトを初期化
    snd = new Sound(req);
    //鳴らす
    channel = snd.play();
}
```

今回は、姿勢を矯正するアプリとしていますが、同じように簡単にバランスゲームを作ることができます。Flashゲームの作り方は多くの書籍やWebサイトで公開されているので、そこにWiiRemoteを使った「ひと工夫」を入れることで、より楽しい作品が作れることでしょう。

## 5.3


## 作例2：ChunChun

制作／原央樹

## WiiRemoteを使ったゲーム

ChunChun は、WiiRemote を入力装置とする Flash ゲームです。WiiRemote とヌンチャクに内蔵されている加速度センサーを使ってスズメの ChunChun を操作し、フルーツを集めるゲームです。

図5-18 ChunChun ゲーム画面



## ChunChun の遊び方

ChunChun は、でぶっちょなスズメなので、自分の小さな羽根だけでは飛ぶことができません。そこで、両手に団扇を持って飛ぶことを考えました。

プレイヤーは WiiRemote とヌンチャクを左右の手に持ちます。右手の WiiRemote を振ると ChunChun は右側の団扇を仰ぎ、左上の方向にふわりと浮きます。逆に左手のヌンチャクを振ると左の団扇を仰ぎ、ChunChun は右上の方向にふわりと移動します。プレイヤーが WiiRemote を操作しなければ ChunChun は落下してしまうので、団扇の仰ぎ具合を絶妙にコントロールし、一定時間内でできるだけたくさんのフルーツを集めてください。

筆者がこのゲームを作るきっかけになったのは、4歳になる息子が団扇で遊んでいるのをみたときです。彼は両手に団扇を持って羽のようにバタバタと動かし、一生懸命空を飛ばうとしていました。

もちろん空を飛ぶことはできないのですが、彼は必死に団扇を動かしジャンプして「空を飛んだ」といいます。実は、僕も幼い頃、団扇で空を飛ばうとしたことがありました。そして、彼と同じようなことを言っていたのだと思います。ということで、親子2代の夢を叶えるべく、このゲームを作成しました。

## ゲームのルール

ChunChunは以下のルールに従って動作を行います。

- ChunChunは重力により落下します
- WiiRemoteを振ると、左上方向に力がかかります
- ヌンチャクを振ると、右上方向に力がかかります

WiiRemoteとヌンチャクをうまく具合にパタパタと振って羽ばたかなければ、すぐに落下してしまい、うまく飛行することができません。

それでは、次項からサンプル制作の流れ、ChunChunのプログラムを見ていきましょう。

## サンプル制作の流れ

以下では、プログラムの該当のパートを引用して解説していきます。全体に関しては、配布するソースコードを参照してください。また、ここではChunChunの操作や動きに関するを中心に解説を進めます。WiiRemoteとの関連性の薄い箇所（あたり判定やスコアの管理など）は、サンプルプログラムのコメントのほうで確認してください。

### WiiRemoteとFlashをつなぐ

WiiRemoteとFlashをつなぐためには、WiiFlashを使用します。前述のように、WiiFlashは、PCにBluetoothを使って接続したWiiRemoteの情報をFlashに受け渡すためのサーバと、その情報をFlashで簡単に扱えるようにしたActionScriptのクラス群で構成されています。

WiiRemoteとPCをBluetoothでつないだら、WiiFlashサーバを立ち上げてください。その後、

次のようにして WiiRemote を Flash で扱えるようにします。

**コード 5-16**   **ActionScript**   初期設定 (net.hara3.chunchun.WiimoteControl.as 抜粋)

```
import org.wiiflash.Wiimote;
import org.wiiflash.events.*;
```

**コード 5-17**   **ActionScript**   Wiimote との接続 (net.hara3.chunchun.WiimoteControl.as 抜粋)

```
private function _wiimoteConnect():void{
    _wiimote = new Wiimote();//WiiRemoteのインスタンスを作成する
    _wiimote.addEventListener(Event.CONNECT, _onWiimoteConnect);
        //WiiRemoteとの接続に成功したら_onWiimoteConnectメソッドを呼び出す
    _wiimote.connect();//WiiRemoteとFlashを接続する
}

//Wiimoteに接続成功したらタイトル画面へ
private function _onWiimoteConnect(e:Event):void {
    SceneManager.getInstance().go("title");
}
```

## WiiRemote の入力を受け取る

WiiRemote からの入力にはさまざまなものがあります。

まずは A ボタン、B ボタンなどの簡単なボタン入力の受け付け方法は次の通りです。

**コード 5-18**   **ActionScript**   ボタン入力の受け付け (net.hara3.chunchun.WiimoteControl.as 抜粋)

```
public function title():void {
    _title_press_A = false;
    _title_press_B = false;

    _wiimote.addEventListener( ButtonEvent.A_PRESS, _onTitleAPress );
        //WiiRemoteのAボタンをPressしたときに、_onTitleAPress()メソッドを実行する
    _wiimote.addEventListener( ButtonEvent.A_RELEASE, _onTitleARelease );
        //WiiRemoteのAボタンをReleaseしたときに、_onTitleARelease()メソッドを実行する
    _wiimote.addEventListener( ButtonEvent.B_PRESS, _onTitleBPress );
        //WiiRemoteのBボタンをPressしたときに、_onTitleBPress()メソッドを実行する
    _wiimote.addEventListener( ButtonEvent.B_RELEASE, _onTitleBRelease );
        //WiiRemoteのBボタンをReleaseしたときに、_onTitleBPress()メソッドを実行する
}
```



では次に、WiiRemoteからの加速度センサーの値を取得してみます。加速度センサーの値を使って、WiiRemoteを振ったときにどれくらいの速さで振ったのかを調べることができます。WiiRemoteが激しく振られたかどうか、マグニチュードを計算して判定します。その計算結果から、WiiRemoteが激しく振られたことがわかったら、ChunChunに浮力を与えます。

コード5-19    **ActionScript** 加速度センサーの値からChunChunに浮力を与える  
(net.hara3.chunchun.WiimoteControl.as 抜粋)

```
public function game(chara_mc:MovieClip):void {

    _chara_mc = chara_mc; //キャラクターを初期化

    _wiimote_motion_1 = new Array(0,0,0);
        //WiiRemoteの加速度を取得するための配列を初期化
    _wiimote_motion_2 = new Array(0, 0, 0);
        //WiiRemoteの加速度を取得するための配列を初期化

    _nunchuk_motion_1 = new Array(0,0,0);
        //WiiRemoteの加速度を取得するための配列を初期化
    _nunchuk_motion_2 = new Array(0, 0, 0);
        //WiiRemoteの加速度を取得するための配列を初期化

    _wiimote.addEventListener(WiimoteEvent.UPDATE, _gameUpdate);
        //WiiRemoteの状態が変化したら、_gameUpdate()メソッドを実行(これをⓂ
        //使って加速度センサーからの情報を監視)
}

//WiiRemoteの状態が変化するたびに呼ばれる
private function _gameUpdate(e:WiimoteEvent):void {
    //リモコン
    _wiimote_motion_1 = _wiimote_motion_2;
        //以前取得した加速度の値を保存しておく
    _wiimote_motion_2 = [e.target.sensorX, e.target.sensorY, e.target.sensorZ];
        //WiiRemoteの加速度の値を配列で取得。X軸・Y軸・Z軸の加速度が取得できる。

    //以前取得した加速度の値と現在取得した加速度の値からマグニチュードを計算(*1)
    var mag:Number = _magnitude(_wiimote_motion_1, _wiimote_motion_2);
    //マグニチュードの値から羽ばたきが十分かどうか判定(*2)
    if (mag > 0.5) {
        _chara_mc.followR(); //マグニチュードが0.5より大きければ、右の羽を振る
    } else {
        _chara_mc.stopR(); //マグニチュードが0.5以下であれば、右の羽を止める
    }

    //ヌンチャク
```

[次ページにつづく](#)

```


_nunchuk_motion_1 = _nunchuk_motion_2; //以前取得した加速度の値を保存しておく
_nunchuk_motion_2 = [e.target.nunchuk.sensorX, e.target.nunchuk.sensorY,
                     e.target.nunchuk.sensorZ];
//ヌンチャクの加速度の値を配列で取得。X軸・Y軸・Z軸の加速度が取得できる。
//以前取得した加速度の値と現在取得した加速度の値からマグニチュードを計算(*1)
mag = _magnitude(_nunchuk_motion_1, _nunchuk_motion_2);
//マグニチュードの値から羽ばたきが十分かどうか判定(*2)
if (mag > .75) {
    _chara_mc.followL(); //マグニチュードが0.75より大きければ、左の羽を振る
}else {
    _chara_mc.stopL(); //マグニチュードが0.75以下であれば、左の羽を止める
}
}

// マグニチュード計算
private function _magnitude(list1:Array, list2:Array):Number {
    var d_x:Number = Math.pow((list1[0] - list2[0]),2);
    var d_y:Number = Math.pow((list1[1] - list2[1]),2);
    var d_z:Number = Math.pow((list1[2] - list2[2]),2);
    return (d_x + d_y + d_z);
}

```

マグニチュード計算では、WiiRemoteがどれだけ激しく振られたかを計算します。  
計算方法は次の図の通りです。

図5-19 マグニチュードの計算方法



#### ・X軸に関して

$dx = \sqrt{(\text{加速度センサーのX方向に現在かかっている速度} - \text{加速度センサーのX方向に少し前にかかっていた速度})^2}$

**・Y軸に関して**

$dy = \sqrt{(\text{加速度センサーのY方向に現在かかっている速度} - \text{加速度センサーのY方向に少し前にかかっていた速度})^2}$

**・Z軸に関して**

$dz = \sqrt{(\text{加速度センサーのZ方向に現在かかっている速度} - \text{加速度センサーのZ方向に少し前にかかっていた速度})^2}$

**・XYZ全方向に関して**

$dxyz = dx + dy + dz$

ただし、WiiRemoteとヌンチャクでは中に入っている加速度センサーの種類が異なるようです。そのため、このサンプルではマグニチュードの判定基準をWiiRemoteでは0.5、ヌンチャクは0.75としています。これは、実際に操作して違和感があるようなら値を調整してみてください。サンプルコードでは、 $\sqrt{}$ （ルート）の計算を行っていません。 $\sqrt{}$ の計算をすると、値が小さくなりすぎ、WiiRemoteの動きがわかりにくかったためです。

**ChunChunの動作**

ChunChunは基本的に自由落下します。WiiRemoteからの羽ばたきを受け付けたときに、浮力を得て移動することができます。

まずは、ChunChunを自由落下する部分を見ていきます。自由落下の計算式は用途によってさまざまな形がありますが、今回は最も簡単なオイラー法を用いた計算を行いたいと思います。

たとえば、

`_G`：重力加速度

`_ax`：X軸方向に常にかかる加速度

`_vy`：物体のY軸方向にかかる力

`_vx`：物体のX軸方向にかかる力

`_dT`：単位時間（どれくらいの時間の精度で計算するか）

としたときの、物体の座標 (x,y) は、

`_vy += _G * _dT;`


```

_vx += _ax * _dT;
x += _vx * _dT;
y += _vy * _dT;

```

となります。

図 5-20 ChunChun の自由落下



これをそのままプログラムに反映します。

コード 5-20 **ActionScript** ChunChun の自由落下 (net.hara3.chunchun.WiimoteControl.as 抜粋)

```

//重力計算用のプロパティ
private var _G:Number = 200;
//重力加速度(Y軸方向に常にかかる加速度)
private var _ax:Number = 0;
//X軸方向に常にかかる加速度(今回はX軸方向に余計な力はいらないので0)
private var _vx:Number = 0; //ChunChunのX方向にかかる力
private var _vy:Number = 0; //ChunChunのY方向にかかる力
private var _dT:Number = 1 / 20; //単位時間(1/フレームレート)

157-169行目
//オイラー法により自由落下運動させる
_vy += _G * _dT; //Y軸の加速度の計算
_vx += _ax * _dT; //X軸の加速度の計算

x += _vx * _dT; //加速度からChunChunのX座標の計算
y += _vy * _dT; //加速度からChunChunのY座標の計算

```

次に、羽ばたいたときに任意の方向に力を加えてみたいと思います。

WiiRemoteを激しく振ったときには、ChunChunの画面に向かって左上方向(たとえば [x:-5, y:-15]) に力が加わるので、その値を羽ばたきの力として `_vx`、`_vy` に加算します。

図5-21 右の羽ばたきの動き




コード5-21 **ActionScript** 右の羽ばたきの動き (net.hara3.chunchun.WiimoteControl.as 抜粋)

```
public function followR():void {
    _wing2_mc.followStart();
    //画面の高い位置の場合は羽ばたき無視
    if (y <= 20) {
        return;
    }

    _vy += -15;
    _vx += -5;
}
```

同様にヌンチャクを激しく振ったときには、ChunChunの画面に向かって右上方向(たとえば [x:5, y:-15]) に力が加わるので、その値を羽ばたきの力として `_vx`、`_vy` に加算します。

図 5-22 左の羽ばたきの動き

コード 5-22 **ActionScript** 左の羽ばたきの動き (net.hara3.chunchun.WiimoteControl.as 抜粋)

```
public function followL():void {
    _wing1_mc.followStart();
    //_wing1_mc.play();

    //画面の高い位置の場合は羽ばたき無視
    if (y <= 20) {
        return;
    }

    _vy += -15;
    _vx += 5;
}
```

もし、ChunChunに加える力が大きすぎたり小さすぎたりして、思ったような操作感が得られない場合は、\_vx、\_vyに加える力を少し調整してみてください。

いかがでしょうか。

ここまでのプログラムにより、FlashがWiiRemoteの羽ばたきを感知し、落下しているChunChunの\_vx、\_vyに力を加え、ChunChunがふわりと浮く感じが表現できたと思います。

あとはさまざまなオブジェクトを配置し、たとえばフルーツを取るとスコアがアップする、というようなゲームとしての仕様を追加実装してみてください。